

The Walker Function

Mikhail D. Mikhailov
Atila P. Silva Freire

The special function $\Xi(\eta)$ (the Walker function) and its derivatives are important for the description of near-wall turbulent flows. This article gives exact expressions for these functions, based on original identities for the hypergeometric functions ${}_1F_1$ and ${}_pF_p$. We also introduce a new initial value problem that generates interpolating functions for $\Xi(\eta)$ and its derivatives.

■ Introduction

The quantitative description of turbulent flows is known to be severely hampered by the extremely rapid variations in the mean and higher-order statistics in the near-wall region. Some very early studies [1, 2, 3] showed that the basic structure of an attached turbulent boundary layer consists of a viscous wall layer, in which the turbulent and laminar stresses are of comparable magnitude, and a defect layer, in which the velocity profile may be expressed in terms of a small perturbation to the external flow solution [4]. Also, [1, 2, 3] showed that this structure naturally leads to a universal velocity solution that has logarithmic behavior and depends on the velocity and length scales based on the friction velocity.

For separated flows, this picture changes dramatically; the friction velocity reduces to zero, so that a new scaling procedure based on the local pressure gradient at the wall needs to be introduced. The logarithmic solution is then reduced to a square-root velocity profile. Aspects of the complex flow structure due to separation are thoroughly discussed in [4].

To incorporate the structure of organized motion into a near-wall analytical model, Walker et al. [5] proposed a time-dependent formulation based on the dynamical features of the wall layer. Following the general description commonly found in the literature, the flow dynamics are considered to be dominated by two features: wall layer streaks and the bursting phenomenon. The streaks are elongated in the flow direction, with length of the order of $1000 \delta^+$ (where $\delta^+ = \nu/u^*$, ν is the kinematic viscosity, and u^* is the friction velocity), and can be observed over a large characteristic time, the quiescent period [6].

Analytical near-wall solutions for turbulent flows are of central significance, since the extreme thinness of the viscous sublayer naturally demands the use of exceptionally fine meshes in the numerical computation of problems of practical relevance. This difficulty can be naturally overcome, provided the local analytical solutions are used to bridge the viscous-dominated region.

In [5], the duration of the quiescent period is assumed to be much larger than the duration of the bursting process, so that solutions for the unsteady Navier–Stokes equations can be given in terms of the flow development between pairs of streaks. The analysis resorts to standard methods for orthogonal functions to arrive at a set of three nonlinear governing equations that have to be solved numerically. In addition, a semi-similarity procedure is proposed, resulting in a solution that includes $\Xi(\eta)$. This function is a similarity solution of the linear diffusion equation that behaves logarithmically for large η and vanishes at $\eta = 0$. Mean velocity profiles are obtained by evaluating a time-average of the solution over the period between bursts, T_B^+ .

In mathematics Ξ is generally used to denote a variant of the Riemann zeta function. Here, no confusion should arise: the function Ξ , as defined in [5], is given by the equation e1 below.

Instantaneous and averaged velocity profiles are presented for four combinations of p^+ (the dimensionless pressure gradient parameter) and T_B^+ in accordance with typically measured values. For constant pressure flow ($p^+ = 0.$, $T_B^+ = 110.2$), the canonical boundary layer structure is well reproduced. Under a favorable pressure gradient ($p^+ = -0.098$, $T_B^+ = 164$), the instantaneous profiles are accelerated as expected. For adverse pressure gradients ($p^+ = 0.11$, $T_B^+ = 29.8$; $p^+ = 0.5$, $T_B^+ = 25$), T_B^+ decreases and, for the latter case, reverse instantaneous flow is observed over the latter part of the cycle.

Here, we develop a new version of the near-wall turbulent velocity profile shown in [5], in particular, of the $\Xi(\eta)$ function. This article introduces the required features of function $\Xi(\eta)$ and its derivatives. In addition, we present some original identities for the hypergeometric functions ${}_1F_1$ and ${}_pF_p$.

■ Identity Involving ${}_1F_1$ and ${}_pF_p$

The function $\Xi(\eta)$ is of particular relevance, being the only known similarity solution of the linear diffusion equation that behaves logarithmically for large η and vanishes at $\eta = 0$ [5].

The $\Xi(\eta)$ function can be defined either in terms of an expansion or an integral, as shown in [5] (equations (44) and (42)). In the latter form, the triple integral can be rearranged to a single integral.

$$\mathbf{e1} = \Xi[\eta] == \frac{e^{-\eta^2}}{4} \sum_{j=1}^{\infty} \frac{2^j \mathbf{d}[j] \eta^{2j+1}}{(2j+1)!!} == \int_0^\eta \int_0^\xi \int_0^t e^{-\xi^2+t^2-x^2} dx dt d\xi$$

$$\Xi[\eta] == \frac{1}{4} e^{-\eta^2} \sum_{j=1}^{\infty} \frac{2^j \eta^{1+2j} \mathbf{d}[j]}{(1+2j)!!} ==$$

$$\int_0^\eta \frac{1}{2} e^{-\xi^2} \xi^2 \text{HypergeometricPFQ}\left[\{1, 1\}, \left\{\frac{3}{2}, 2\right\}, \xi^2\right] d\xi$$

The coefficients $\mathbf{d}[j]$ are defined by a recurrence relation.

$$\mathbf{e2} = \{\mathbf{d}[j] == \mathbf{d}[j-1] + 1/j, \mathbf{d}[1] == 1\};$$

The sum in $\mathbf{e1}$ is uniformly convergent for all η . This is particularly useful in the evaluation of $\Xi(\eta)$ for small to moderate values of η [5].

This solves the recurrence equation.

$$\mathbf{e3} = \text{RSolve}[\mathbf{e2}, \mathbf{d}[j], j][[1, 1]]$$

$$\mathbf{d}[j] \rightarrow \text{EulerGamma} + \text{PolyGamma}[0, 1 + j]$$

`FullSimplify` shows that the coefficients $\mathbf{d}[j]$ are the harmonic numbers.

$$\mathbf{e4} = \text{FullSimplify}[\text{RSolve}[\mathbf{e2}, \mathbf{d}[j], j][[1, 1]]]$$

$$\mathbf{d}[j] \rightarrow \text{HarmonicNumber}[j]$$

After substituting equation $\mathbf{e4}$ into equation $\mathbf{e1}$, the sum turns out to be a hypergeometric function.

$$\mathbf{e5} = \mathbf{e1} /. \mathbf{e4}$$

$$\Xi[\eta] == \frac{1}{4} e^{-\eta^2} \eta \text{Hypergeometric1F1}^{(1,0,0)}\left[1, \frac{3}{2}, \eta^2\right] ==$$

$$\int_0^\eta \frac{1}{2} e^{-\xi^2} \xi^2 \text{HypergeometricPFQ}\left[\{1, 1\}, \left\{\frac{3}{2}, 2\right\}, \xi^2\right] d\xi$$

This is the second term of e5.

e5[[2]]

$$\frac{1}{4} e^{-\eta^2} \eta \text{Hypergeometric1F1}^{(1,0,0)} \left[1, \frac{3}{2}, \eta^2 \right]$$

This is difficult to calculate for large numerical values of η , because $e^{-\eta^2}$ is extremely small, while the hypergeometric function is extremely large. To remove $e^{-\eta^2}$ from e5[[2]], we use the following identity.

$$\begin{aligned} \text{FullSimplify}[\\ \mathbf{e6} = \text{Hypergeometric1F1}[\alpha, \gamma, \eta^2] = \\ e^{\eta^2} \text{Hypergeometric1F1}[\gamma - \alpha, \gamma, -\eta^2] \end{aligned}$$

True

Take the derivative of e6 with respect to α for $\alpha = 1$ and $\gamma = 3/2$.

$$\mathbf{e7} = \text{Rule@@D}[\mathbf{e6}, \alpha] /. \{\alpha \rightarrow 1, \gamma \rightarrow 3/2\}$$

$$\begin{aligned} \text{Hypergeometric1F1}^{(1,0,0)} \left[1, \frac{3}{2}, \eta^2 \right] \rightarrow \\ -e^{\eta^2} \text{Hypergeometric1F1}^{(1,0,0)} \left[\frac{1}{2}, \frac{3}{2}, -\eta^2 \right] \end{aligned}$$

Use e7 to substitute into e5.

$$\mathbf{e8} = \mathbf{e5} /. \mathbf{e7}$$

$$\begin{aligned} \mathbb{E}[\eta] = -\frac{1}{4} \eta \text{Hypergeometric1F1}^{(1,0,0)} \left[\frac{1}{2}, \frac{3}{2}, -\eta^2 \right] = \\ \int_0^\eta \frac{1}{2} e^{-\xi^2} \xi^2 \text{HypergeometricPFQ} \left[\{1, 1\}, \left\{ \frac{3}{2}, 2 \right\}, \xi^2 \right] d\xi \end{aligned}$$

To verify that e8 is an identity, we evaluate its right-hand side for $\eta = 6/5$ with 100 digits of precision.

Rest[e8]

$$\begin{aligned} -\frac{1}{4} \eta \text{Hypergeometric1F1}^{(1,0,0)} \left[\frac{1}{2}, \frac{3}{2}, -\eta^2 \right] = \\ \int_0^\eta \frac{1}{2} e^{-\xi^2} \xi^2 \text{HypergeometricPFQ} \left[\{1, 1\}, \left\{ \frac{3}{2}, 2 \right\}, \xi^2 \right] d\xi \end{aligned}$$

```
N[% /. η → 6 / 5, 100]
```

```
True
```

For $\Xi'[\eta]$ and $\Xi''[\eta]$, *Mathematica* 8 gives the following.

```
e9 = D[Drop[e8, -1], η]
```

$$\Xi'[\eta] = -\frac{1}{4} \text{Hypergeometric1F1}^{(1,0,0)}\left[\frac{1}{2}, \frac{3}{2}, -\eta^2\right] + \frac{1}{2} \eta^2 \text{Hypergeometric1F1}^{(1,0,1)}\left[\frac{1}{2}, \frac{3}{2}, -\eta^2\right]$$

```
e10 = D[e9, η]
```

$$\Xi''[\eta] = \frac{3}{2} \eta \text{Hypergeometric1F1}^{(1,0,1)}\left[\frac{1}{2}, \frac{3}{2}, -\eta^2\right] - \eta^3 \text{Hypergeometric1F1}^{(1,0,2)}\left[\frac{1}{2}, \frac{3}{2}, -\eta^2\right]$$

Mathematica 8 can evaluate **e9** and **e10** for numerical values of η . The following rule generates derivatives of order n that earlier versions of *Mathematica* can evaluate.

```
WalkerDerivative[x_, n_: 0] :=  
D[Ξ[x], {x, n}] ==  
Expand[  
  Simplify[  
    PowerExpand[  
      D[- x / 4 Hypergeometric1F1[C, 3 / 2, -x^2], {x, n}, C] /.  
      C -> 1 / 2]]]
```

For $\Xi'[\eta]$ and $\Xi''[\eta]$, the function `WalkerDerivative` gives the following.

```
e11 = WalkerDerivative[η, 1]
```

$$\Xi'[\eta] = -\frac{e^{-\eta^2}}{2} + \frac{\sqrt{\pi} \text{Erf}[\eta]}{4\eta} - \frac{1}{4} \text{Hypergeometric1F1}^{(1,0,0)}\left[\frac{1}{2}, \frac{3}{2}, -\eta^2\right] + \frac{1}{6} \eta^2 \text{Hypergeometric1F1}^{(1,0,0)}\left[\frac{3}{2}, \frac{5}{2}, -\eta^2\right]$$

e12 = WalkerDerivative[η , 2]

$$\begin{aligned} \Xi''[\eta] = & \frac{e^{-\eta^2}}{2\eta} + \frac{4}{3} e^{-\eta^2} \eta - \frac{\sqrt{\pi} \operatorname{Erf}[\eta]}{4\eta^2} + \\ & \frac{1}{2} \eta \operatorname{Hypergeometric1F1}^{(1,0,0)}\left[\frac{3}{2}, \frac{5}{2}, -\eta^2\right] - \\ & \frac{1}{5} \eta^3 \operatorname{Hypergeometric1F1}^{(1,0,0)}\left[\frac{5}{2}, \frac{7}{2}, -\eta^2\right] \end{aligned}$$

To verify that the right-hand sides of e9 and e11 are equal and that the right-hand sides of e10 and e12 are equal, we evaluate them for $\eta = 6/5$ with 100 digits of precision.

N[{Last[e9] == Last[e11], Last[e10] == Last[e12]} /. $\eta \rightarrow 6/5$, 100]

{True, True}

■ Initial Value Problem

To derive the initial value problem for $\Xi[\eta]$, we use e8 and its derivatives.

e13 = Drop[e8, {2}]

$$\Xi[\eta] = \int_0^\eta \frac{1}{2} e^{-\xi^2} \xi^2 \operatorname{HypergeometricPFQ}\left[\{1, 1\}, \left\{\frac{3}{2}, 2\right\}, \xi^2\right] d\xi$$

e14 = D[e13, η]

$$\Xi'[\eta] = \frac{1}{2} e^{-\eta^2} \eta^2 \operatorname{HypergeometricPFQ}\left[\{1, 1\}, \left\{\frac{3}{2}, 2\right\}, \eta^2\right]$$

e15 = Expand[D[e14, η]]

$$\begin{aligned} \Xi''[\eta] = & \\ & \frac{1}{2} \sqrt{\pi} \operatorname{Erf}[\eta] - e^{-\eta^2} \eta^3 \operatorname{HypergeometricPFQ}\left[\{1, 1\}, \left\{\frac{3}{2}, 2\right\}, \eta^2\right] \end{aligned}$$

To obtain the desired initial value problem, we eliminate from e14 and e15 the hypergeometric function and replace η by 0 in e13 and e14.


```
e16 = Prepend[{e13, e14} /.  $\eta \rightarrow 0$ ,
  Equal @@
  Expand[Solve[Eliminate[{e14, e15}, e14[[2, 4]]],  $\Xi''[\eta]$ ][[
    1, 1]]]
```

$$\left\{ \Xi''[\eta] = \frac{1}{2} \sqrt{\pi} \operatorname{Erf}[\eta] - 2 \eta \Xi'[\eta], \Xi[0] = 0, \Xi'[0] = 0 \right\}$$

The resulting ordinary differential equation is equivalent to equations (40) and (41) in [5]. The condition $\Xi[0] = 0$ is also given in [5]. However, the condition $\Xi'[0] = 0$ is missing in [5], where no attempt is made to obtain a numerical solution.

We plot the function $\Xi[\eta]$ (blue) and its derivative $\Xi'[\eta]$ (purple).

```
Manipulate[
  Plot[Evaluate[{NWalker[x], NDWalker[x]}], {x, -10, 10},
  Frame  $\rightarrow$  True, PlotRange  $\rightarrow$  All, Axes  $\rightarrow$  False,
  FrameLabel  $\rightarrow$  {" $\eta$ ", Row[{" $\Xi(\eta)$ ", " $\Xi'(\eta)$ "},
  Style[Row[{" $\Xi$ ", NumberForm[X, {10, 3}], "} = ",
  NumberForm[NWalker[X], {10, 3}], ",  $\Xi'$ ",
  NumberForm[X, {10, 3}], "} = ",
  NumberForm[NDWalker[X], {10, 3}]]], 12}},
  Epilog  $\rightarrow$  {Red, PointSize[.02], Point[{X, NWalker[X]}],
  Green, Point[{X, NDWalker[X]}]},
  {{X, 0, " $\eta$ "}, -10, 10, 0.001, Appearance  $\rightarrow$  "Labeled"},
  Initialization  $\rightarrow$ 
  {NWalker[x_] := If[x < 0, -NWal[-x], NWal[x]];
  NDWalker[x_] := If[x < 0, NDWal[-x], NDWal[x]];
  {NWal, NDWal} = Map[Last,
  First[
  NDSolve[{Z'[x] == Y[x],
  Y'[x] == Sqrt[Pi] Erf[x] / 2 - 2 x Y[x], Z[0] == 0,
  Y[0] == 0}, {Z, Y}, {x, 0, 10}]]],
  ControlPlacement  $\rightarrow$  Bottom]
```



■ Conclusion

Exact expressions for the Walker function $\Xi(\eta)$ and its derivatives are obtained. From a computational point of view, an important result is the specification of the initial value problem, for it allows extremely fast interpolating functions to be generated. From a mathematical point of view, some new identities between the two hypergeometric functions ${}_1F_1$ and ${}_pF_p$ are presented. These identities were revealed by the built-in *Mathematica* functions `RSolve`, `Sum`, and `FullSimplify`. Of possible interest is the application of the techniques to discover new identities and to the problems described in the book by Petkovšek et al. [7].

■ Acknowledgments

The first author is thankful to the Brazilian National Research Council (CNPq) and the Brazilian National Institute of Standards (INMETRO) for their financial support of this research through a CNPq Research Fellowship—Edital MCT/CNPq/Inmetro n° 059/2010—PROMETRO Processo N° 563061/2010–3.

The second author is grateful to the Brazilian National Research Council (CNPq) for the award of a Research Fellowship (Grant No. 303982/2009–8). The work was financially supported by the Rio de Janeiro Research Foundation (FAPERJ) through Grant E–26/170.005/2008.

■ References

- [1] L. Prandtl, "Über die ausgebildete Turbulenz," *ZAMM: Journal of Applied Mathematics and Mechanics/Zeitschrift für Angewandte Mathematik und Mechanik*, **5**, 1925 pp. 136–139.
- [2] T. von Kármán, "Mechanische Aehnlichkeit und Turbulenz," *Proceedings of the Third International Congress for Applied Mechanics*, Vol. 1, Stockholm, 1930 (C. W. Oseen and W. Weibull, eds.), Stockholm: AB. Sveriges Litografiska Tryckerier, 1931 pp. 79-93 (in German).
- [3] C. B. Millikan, "A Critical Discussion of Turbulent Flow in Channels and Circular Tubes," *Proceedings of the 5th International Congress for Applied Mechanics*, Cambridge, MA, 1938, New York: Wiley, 1939 pp. 386–392.
- [4] J. B. R. Loureiro, A. P. Silva Freire, "Scaling of Turbulent Separating Flows," *International Journal of Engineering Science*, **49**(5), 2011 pp. 397–410. doi:10.1016/j.ijengsci.2010.12.001.
- [5] J. D. A. Walker, D. E. Abbott, R. K. Scharnhorst, and G. G. Weigand, "Wall-Layer Model for the Velocity Profile in Turbulent Flows," *American Institute of Aeronautics and Astronautics Journal*, **27**(2), 1989 pp. 140–149.
- [6] S. J. Kline, W. C. Reynolds, F. A. Schraub, and P. W. Rundstadler, "The Structure of Turbulent Boundary Layers," *Journal of Fluid Mechanics*, **30**(4), 1967 pp. 741–773. journals.cambridge.org/abstract_S0022112067001740.
- [7] M. Petkovšek, H. S. Wilf, D. Zeilberger, *A=B*, Wellesley, MA: A. K. Peters, 1996. www.math.upenn.edu/%7Ewilf/AeqB.html.

M. D. Mikhailov and A. P. Silva Freire, "The Walker Function," *The Mathematica Journal*, 2012. dx.doi.org/doi:10.3888/tmj.14-11.

About the Authors

M. D. Mikhailov

*INMETRO—Divisão de Metrologia em Dinâmica de Fluidos
Dimci/Dinam, Prédio 3, Av. Nossa Senhora das Graças 50
Duque de Caxias, 25250-020
Rio de Janeiro, Brasil
m.d.mikhailov@gmail.com*

Atila P. Silva Freire

*Mechanical Engineering Program (PEM/COPPE/UFRJ)
C.P. 68503, 21941-972
Rio de Janeiro, Brazil
atila@mecanica.coppe.ufrj.br*